# $Id$
# Maintainer: Evangelos Foutras <evangelos@foutrelis.com>
# Contributor: tobias <tobias funnychar archlinux.org>

pkgname=gtk-xfce-engine
pkgver=2.10.1
pkgrel=1
pkgdesc="Xfce Gtk+-2.0 engine and themes"
arch=('i686' 'x86_64')
url="http://www.xfce.org/"
license=('GPL2')
groups=('xfce4')
depends=('gtk2')
provides=("gtk2-xfce-engine=$pkgver-$pkgrel")
conflicts=('gtk2-xfce-engine')
replaces=('gtk2-xfce-engine' 'gtk3-xfce-engine')
source=(http://archive.xfce.org/src/xfce/$pkgname/${pkgver%.*}/$pkgname-$pkgver.tar.bz2)
sha256sums=('4a92910205881f707f2cc8d3c9b00a95feb58eb9084d932c841f3ed027f7063d')

build() {
  cd "$srcdir/$pkgname-$pkgver"

  ./configure \
    --prefix=/usr \
    --sysconfdir=/etc \
    --libexecdir=/usr/lib \
    --localstatedir=/var \
    --disable-static \
    --disable-debug
  make
}

package() {
  cd "$srcdir/$pkgname-$pkgver"
  make DESTDIR="$pkgdir" install
}

# vim:set ts=2 sw=2 et:
